```
(* Problem 4.8 *)

(* VELOCITY ANALYSIS - input angle phi *)

Apply[Clear,Names["Global`*"]];
Off[General::spell];
Off[General::spell1];

n = 100 ; (* rpm *)
omega = n*N[Pi]/30 ; (* rad/s *)

initdata = {AB->0.12, AC->0.06, BD->0.24, DE->0.33,EF->0.19,La->.3, Lb->.07, phi[t]-
>N[Pi]150/180, phi'[t]->omega, phi''[t]->0};


(* Position of joint A *)
xA = yA = 0;

(* Position of joint C *)
xC = -AC ;
yC = 0 ;

(* Position of joint F *)
xF = La ;
yF = Lb ;

(* Position of joint B *)
xB = AB Cos[phi[t]] ;
yB = AB Sin[phi[t]] ;

Print["xB = ", xB ," = ", xB/.initdata, " m" ];
Print["yB = ", yB ," = ", yB/.initdata, " m" ];

(* Linear velocity of joint B *)
vBx = D[xB,t];
vBy = D[yB,t];
Print["vBx = ", vBx ," = ", vBx/.initdata, " m/s" ];
Print["vBy = ", vBy ," = ", vBy/.initdata, " m/s" ];

(* Linear acceleration of joint B *)
aBx = D[vBx,t];
aBy = D[vBy,t];
Print["aBx = ", aBx ," = ", aBx/.initdata, " m/s^2" ];
Print["aBy = ", aBy ," = ", aBy/.initdata, " m/s^2" ];

(* Position of joint D *)
eqnD1 = ( xDsol - xB )^2 + ( yDsol - yB )^2 - BD^2 == 0 ;
eqnD2 = ( yDsol - yB ) / ( xDsol - xB ) == ( yB - yC ) / ( xB - xC );
solutionD = Solve [ { eqnD1 , eqnD2 } , { xDsol , yDsol } ];
(* Two solutions for D *)
xD1 = xDsol /. solutionD[[1]];
yD1 = yDsol /. solutionD[[1]];
xD2 = xDsol /. solutionD[[2]];
yD2 = yDsol /. solutionD[[2]];

If [ (yD1/.initdata) <= 0 , xD = xD1 ; yD = yD1 , xD = xD2 ; yD=yD2 ] ;

Print["xD = ",xD/.initdata," m"];
Print["yD = ",yD/.initdata," m"];

(* Linear velocity of joint D *)
vDx = D[xD,t];
```

```
vDy = D[yD,t];
Print["vDx = ", vDx/.initdata, " m/s" ];
Print["vDy = ", vDy/.initdata, " m/s" ];

(* Linear acceleration of joint D *)
aDx = D[vDx,t];
aDy = D[vDy,t];
Print["aDx = ", aDx/.initdata, " m/s^2" ];
Print["aDy = ", aDy/.initdata, " m/s^2" ];

(* Angular velocity and acceleration of the link BD (link 2)*)
phi2 = ArcTan[ (yB-yC)/(xB-xC) ] ;
omega2 = D[ phi2 , t ] ;
alpha2 = D[ omega2, t ] ;
Print["phi2 = phi3 = ", phi2/.initdata , " rad = ",(phi2/.initdata)*180/N[Pi]," deg" ];
Print["omega2 = omega3 = ", omega2/.initdata , " rad/s" ];
Print["alpha2 = alpha3 = ", alpha2/.initdata , " rad/s^2" ];

(* Position of joint E *)
eqnE1 = ( xEsol - xD )^2 + ( yEsol - yD )^2 - DE^2 == 0 ;
eqnE2 = ( xEsol - xF )^2 + ( yEsol - yF )^2 - EF^2 == 0 ;
solutionE = Solve [ { eqnE1 , eqnE2 } , { xEsol , yEsol } ];
(* Two solutions for E *)
xE1 = xEsol /. solutionE[[1]];
yE1 = yEsol /. solutionE[[1]];
xE2 = xEsol /. solutionE[[2]];
yE2 = yEsol /. solutionE[[2]];
If [ (yE1/.initdata) <= 0 , xE = xE1 ; yE = yE1 , xE = xE2 ; yE=yE2 ] ;
Print["xE = ",xE/.initdata," m"];
Print["yE = ",yE/.initdata," m"];

(* Linear velocity of joint E *)
vEx = D[xE,t];
vEy = D[yE,t];
Print["vEx = ", vEx/.initdata, " m/s" ];
Print["vEy = ", vEy/.initdata, " m/s" ];

(* Linear acceleration of joint D *)
aEx = D[vEx,t];
aEy = D[vEy,t];
Print["aEx = ", aEx/.initdata, " m/s^2" ];
Print["aEy = ", aEy/.initdata, " m/s^2" ];

(* Angular velocity and acceleration of the link DE (link 4)*)
phi4 = ArcTan[ (yD-yE)/(xD-xE) ] ;
omega4 = D[ phi4 , t ] ;
alpha4 = D[ omega4, t ] ;
Print["phi4 = ", phi4/.initdata , " rad = ",(phi4/.initdata)*180/N[Pi]," deg" ];
Print["omega4 = ", omega4/.initdata , " rad/s" ];
Print["alpha4 = ", alpha4/.initdata , " rad/s^2" ];

(* Angular velocity and acceleration of the link EF (link 5)*)
phi5 = ArcTan[ (yE-yF)/(xE-xF) ] ;
omega5 = D[ phi5 , t ] ;
alpha5 = D[ omega5, t ] ;
Print["phi5 = ", phi5/.initdata , " rad = ",(phi5/.initdata)*180/N[Pi]," deg" ];
Print["omega5 = ", omega5/.initdata , " rad/s" ];
Print["alpha5 = ", alpha5/.initdata , " rad/s^2" ];

rB={xB/.initdata,yB/.initdata,0};
rC={xC/.initdata,yC/.initdata,0};
rBC=rC-rB;
Print["rB = ", rB, " m" ];
Print["rC = ", rC, " m" ];
Print["rBC = rC - rB = ", rBC, " m" ];
```

```
vB={vBx/.initdata,vBy/.initdata,0};
aB={aBx/.initdata,aBy/.initdata,0};
Print["vB = ", vB , " m/s" ];
Print["aB = ", aB , " m/s" ];
ω2={0,0,omega2/.initdata};
α2={0,0,alpha2/.initdata};
Print["ω2 = ", ω2, " rad/s" ];
Print["α2 = ", α2, " rad/s^2" ];
vC2=vB+Cross[ω2,rBC];
Print["vC2 = vB + ω2 x rBC = ",vC2, " m/s" ];
aC2=aB+Cross[α2,rBC]-ω2.ω2 rBC;
Print["aC2 = aB + α2 x rBC - ω2^2 rBC = ", aC2, " m/s^2" ];
vC32=-vC2;
Print["vC32 = vC3 - vC2 = ",vC32, " m/s" ];
aC32cor=2 Cross[ω2,vC32];
Print["aC32cor = 2 ω2 x vC32 = ",aC32cor, " m/s^2" ];
aC32=-aC2-aC32cor;
Print["aC32 = aC3 - aC2 - aC32cor = ",aC32, " m/s^2" ];
rD={xD/.initdata,yD/.initdata,0};
rBD=rD-rB;
Print["rD = ", rD, " m" ];
Print["rBD = rD - rB = ", rBD, " m" ];
vD=vB+Cross[ω2,rBD];
Print["vD = vB + ω2 x rBD = ",vD, " m/s" ];
aD=aB+Cross[α2,rBD]-ω2.ω2 rBD;
Print["aD = aB + α2 x rBD - ω2^2 rBD = ", aD, " m/s^2" ];
```

```
xB = AB Cos[phi[t]] = -0.103923 m

yB = AB Sin[phi[t]] = 0.06 m

vBx = -AB Sin[phi[t]] phi'[t] = -0.628319 m/s

vBy = AB Cos[phi[t]] phi'[t] = -1.08828 m/s

aBx = -AB Cos[phi[t]] phi'[t]^2 - AB Sin[phi[t]] phi''[t] = 11.3964 m/s^2

aBy = -AB Sin[phi[t]] phi'[t]^2 + AB Cos[phi[t]] phi''[t] = -6.57974 m/s^2

xD = 0.0378427 m

yD = -0.133656 m

vDx = 2.36622 m/s

vDy = 1.10387 m/s

aDx = 4.50619 m/s^2

aDy = 59.4962 m/s^2

phi2 = phi3 = -0.938882 rad = -53.794 deg

omega2 = omega3 = 15.4632 rad/s

alpha2 = alpha3 = 139.461 rad/s^2

xE = 0.366833 m

yE = -0.107858 m

vEx = 2.38257 m/s
```

```
vEy = 0.895288 m/s

aEx = 6.00543 m/s^2

aEy = 38.6801 m/s^2

phi4 = 0.078255 rad = 4.48368 deg

omega4 = -0.634015 rad/s

alpha4 = -63.2412 rad/s^2

phi5 = -1.21135 rad = -69.4055 deg

omega5 = 13.3959 rad/s

alpha5 = 101.197 rad/s^2

rB = {-0.103923, 0.06, 0} m

rC = {-0.06, 0, 0} m

rBC = rC - rB = {0.043923, -0.06, 0} m

vB = {-0.628319, -1.08828, 0} m/s

aB = {11.3964, -6.57974, 0} m/s

ω2 = {0, 0, 15.4632} rad/s

α2 = {0, 0, 139.461} rad/s^2

vC2 = vB + ω2 x rBC = {0.299474, -0.409089, 0.} m/s

aC2 = aB + α2 x rBC - ω2^2 rBC = {9.26164, 13.8925, 0.} m/s^2

vC32 = vC3 - vC2 = {-0.299474, 0.409089, 0.} m/s

aC32cor = 2 ω2 x vC32 = {-12.6516, -9.26164, 0.} m/s^2

aC32 = aC3 - aC2 - aC32cor = {3.39, -4.63082, 0.} m/s^2

rD = {0.0378427, -0.133656, 0} m

rBD = rD - rB = {0.141766, -0.193656, 0} m

vD = vB + ω2 x rBD = {2.36622, 1.10387, 0.} m/s

aD = aB + α2 x rBD - ω2^2 rBD = {4.50619, 59.4962, 0.} m/s^2
```